

Paper 425-2013

Price- and Cross-Price Elasticity Estimation using SAS[®]

Dawit Mulugeta, Jason Greenfield, Tison Bolen and Lisa Conley, Cardinal Health,

Pricing Analytics Team, Dublin, Ohio 43017, USA

ABSTRACT

The relationship between price and demand (quantity) has been the subject of extensive studies across many product categories, regions, and stores. Elasticity estimates have also been used to improve pricing strategies and price optimization efforts, promotions, product offers, and various marketing programs. This presentation demonstrates how to compute item-level price and cross-price elasticity values for two products with and without promotions. We used the midpoint formula, the OLS linear model, and the log-log model to measure demand response to change in price using six-month transaction-level data. Limitations and prospects of the methods used are discussed. The inclusion of promotions and prices of other products as covariates provides a better understanding of the dynamics of price-demand relationships.

INTRODUCTION

The price elasticity of demand measures the responsiveness of consumers to change in the price of a product [5, 9, 14]. It is commonly computed as the percentage change in demand or quantity divided by the percentage change in price. Since the development of the concept of price elasticity of demand from marginal utility theory in 1890 [12], price elasticity estimation has long been the subject of many studies, and takes prominent place in many econometrics text books, several publications, market research and business consultation efforts.

Estimation of price elasticity serves many purposes. Once the demand response to price is known, it is possible to implement store- or customer-specific promotion expenditure and pricing strategies including choice of regular prices, magnitude of discounts, product bundling, product positioning and pricing of private labels. Price elasticity estimation have been the subject of many studies for various product groups including gasoline [10], beef [7], timber [17], cigarettes [1], alcoholic beverages [16], online transaction data [11], sales of digital scientific information [13], a range of postal products [15] and several consumer good items [3, 4, 5, 6]. Variability of price elasticities were measured across store chains [2, 9, 10], store and national brands [9], regions [16, 17], time periods [3, 6, 16] and stages of product life cycle [15]. Also price elasticity estimations are powerful tools to optimize prices for improved revenue and profits [13, 14] and develop competitive strategy analysis and market power indices [2].

In this presentation we will show a quick way of measuring item-level elasticity using SAS[®]. Within the framework of a free market where competition of goods and prices occur, we demonstrate the impacts of factors that influence consumer demand. We will use two products and relevant covariates to estimate own and cross elasticities. We will also provide ways to interpret the results.

METHODS

For this analysis, two pharmaceutical drugs, product A and product B were selected. We initially assume these drugs to be perfectly substitutable, are at similar stage in their lifecycle, are subject to similar competition and cost dynamics in the market, and were available for sale during the course of the study with no inventory or supply-chain constraints. We use six months artificial weekly data to estimate own and cross elasticities. Two covariates: promotion 1 (telephone call) and promotion 2 (web ad) were included to assess the impacts of some marketing variables on elasticity. There are many channels of promotion at Cardinal Health, a B2B environment, amongst which telephone calls and web ads are the most common. Telephone calls are made to customers about price reduction and value offers of selected products. In addition, web-based advertisements on price and product launches are occasionally done to selected customers. The two promotional variables have a dummy variable of 1 when promotion occurs, otherwise the dummy variable equals zero.

We first measured own elasticity using mid-point elasticity estimation method [5, 14] as shown in the following equation:

$$E_A = \% \Delta Q_A / \% \Delta P_A \quad (1)$$

Where:

E_A = Own price elasticity

$(\% \Delta Q_A)$ = Percent change in quantity (Q) of product A computed as

$$[(Q_{A(w)} - Q_{A(w-1)}) / (Q_{A(w-1)} + Q_{A(w)}) / 2]$$

$(\% \Delta P_A)$ = Percent change in price (P) of product A computed as

$$[(P_{A(w)} - P_{A(w-1)}) / (P_{A(w-1)} + P_{A(w)}) / 2]$$

w and $(w-1)$ refer to current and previous weeks, respectively.

Upon algebraic rearrangement the above equation (1) can be expressed as follows:

$$E_A = (\Delta Q_A / \Delta P_A) * (\bar{P}_A / \bar{Q}_A) \quad (2)$$

Where:

(ΔQ_A) = Change in quantity of product A computed as $(Q_{A(w)} - Q_{A(w-1)})$

(ΔP_A) = Change in price of product A = $(P_{A(w)} - P_{A(w-1)})$

\bar{P}_A and \bar{Q}_A are the average price and quantity of product A, respectively.

Cross elasticity was computed as:

$$CE_{A,B} = \% \Delta Q_A / \% \Delta P_B \quad (3)$$

Where

$(\% \Delta Q_A)$ as a described above in (1)

$(\% \Delta P_B)$ = Percent change in price (P) of product B computed as

$$[(P_{B(w)} - P_{B(w-1)}) / (P_{B(w-1)} + P_{B(w)})/2]$$

Own and cross elasticity estimates were computed for each of the 26 weeks. Both values were not estimated when prices remain the same in adjacent weeks.

Linear and log-log demand functions were compared to model the relationship between quantity of product and price of A as follows:

$$Q_A = \beta_0 + \beta_1 * P_A + \varepsilon \quad (4)$$

Where:

Elasticity was computed as:

$$\beta_1 * (\bar{P}_A / \bar{Q}_A)$$

$$\text{Log } Q_A = \beta_0 + \beta_1 * \text{log } P_A \quad (5)$$

Where:

β_1 is own elasticity of product A.

Similarly a standard log-log demand function was estimated using ordinary least squares (OLS) regression as follows:

$$\text{log } Q_A = \beta_0 + \beta_1 * \text{log } P_A + \beta_2 * \text{log } P_B + \beta_3 * \text{Promo}_1 + \beta_4 * \text{Promo}_2 + \beta_5 * (\text{log } P_A * \text{Promo}_1) + \beta_6 * (\text{log } P_A * \text{Promo}_2) + \varepsilon \quad (6)$$


Where:

$\text{log } Q_A$, $\text{log } P_A$, and $\text{log } P_B$ represent the log values of quantity of product A, log values of prices of products A and B, respectively; Promo_1 and Promo_2 represent promotion1 (phone call) and promotion2 (web ad), respectively; β_0 is the intercept; β_1 to β_6 are model parameter estimates where β_1 and β_2 are own and cross price elasticities of Product A, respectively.

RESULTS AND DISCUSSION

All possible relationships between price and demand are shown in a 3 X 3 grid (Figure 1). When market prices are changing demand may go up or down or may remain the same.


Figure 1 Possible relationships between price and demand (quantity).



Sectors 1 and 9 are cases in which price and demand follow the same directional changes. This is likely when other marketing forces such as competition, inclement weather, introduction of new outlet or product occur. Sectors 2, 5 and 8 happen when demand remains the same irrespective of price changes; that means no sales occurred. Responses shown in sector 2 and 8 are commonly said to have a perfectly inelastic demand. Sectors 4 and 6 are cases where demand goes up or down while price remain the same. Such kind of a response is commonly called a perfectly elastic demand.

Among the nine possible demand responses to price changes (Figure 1), only two (sectors 3 and 7) lend themselves to elasticity estimation, and such responses are the subject of many demand-price relation studies [2, 3, 4, 5, 6, 15, 16, 17] including this demonstration. Demand response in sectors 3 and 7 where price and demand moves the opposite direction can take up different forms and values [5]. Although price-demand relationships can assume various functional relationships (linear, quadratic, logistic, etc.), the focus in this presentation is on the linear form. A hypothetical example is presented in Figure 2 where the relationship between price and demand of a single product is shown using units and price values that range from 0 to 10.

Figure 2 Elasticity gradients along a linear price-demand curve.



Whether elasticity is estimated using the mid-point formula or the regression demand-response models shown in many of the reference papers, elasticity values in sectors 3 and 7 of Figure 1 can have values of 0 to negative infinity as shown in Figure 2.

For the same sectors of 3 and 7 of Figure 1, when elasticity values are less than zero and greater than -1 the demand response is referred as inelastic. This is the case when relative price change brings a corresponding lesser percentage in demand. If the value equals -1, then the response is referred as unit (unitarily) elastic demand. This is typical when percent change in price brings an equal percent change in demand. When elasticity values are less than -1 then the demand response is termed as elastic, the more the number is negative the more the response is elastic.

Figure 2 shows that a move from point A to point B of the elastic and inelastic portions of the curve have resulted in a single unit of price and demand changes on both sides. However, a reduction of price from \$9 to \$8 in the elastic portion of the curve brought a revenue (price * demand) increase of 78%, whereas a similar price change from \$2 to \$1 incurred a -44% change in revenue. Thus the slope of such a line and the proportion of the line shared between the elastic and the inelastic portions of the curve impacts the revenue derived from price changes.

The data used for this analysis are shown in Table 1. The quantity of product A and the prices of both products, flags of two promotions as well as own and cross-price elasticity estimates (based on two weeks mid-point estimation) are shown.

Table 1 Price, quantity and promotion data of product A; price of product B with own and cross-elasticity values estimated using the two-point estimation method.

Week	Product A Quantity	Product A Price	Product B Price	Promotion 1	Promotion 2	Own Elasticity	Cross Elasticity
1	70	\$7.50	\$7.00	0	0	*	*
2	75	\$7.30	\$7.15	0	0	-2.55	3.25
3	75	\$7.25	\$7.15	0	0	0.00	*
4	78	\$7.25	\$7.15	0	0	*	*
5	100	\$6.50	\$7.20	1	0	-2.27	35.47
6	120	\$6.50	\$7.20	1	0	*	*
7	95	\$6.50	\$7.25	1	0	*	-33.60
8	115	\$6.50	\$7.25	1	0	*	*
9	78	\$7.05	\$7.25	0	0	-4.72	*
10	85	\$7.05	\$7.25	0	1	*	*
11	88	\$7.05	\$7.30	0	1	*	5.05
12	84	\$7.05	\$7.30	0	1	*	*
13	90	\$7.05	\$7.30	0	1	*	*
14	86	\$7.05	\$7.40	0	0	*	-3.34
15	95	\$7.00	\$7.40	0	0	-13.97	*
16	88	\$7.00	\$7.40	0	0	*	*
17	105	\$6.90	\$7.40	0	0	-12.24	*
18	100	\$6.90	\$7.45	0	1	*	-7.24
19	108	\$6.90	\$7.45	0	1	*	*
20	95	\$6.90	\$7.50	0	1	*	-19.15
21	98	\$6.90	\$7.50	0	1	*	*
22	130	\$6.45	\$7.50	1	0	-4.16	*
23	125	\$6.45	\$7.70	1	0	*	-1.49
24	133	\$6.45	\$7.75	1	0	*	9.58
25	128	\$6.45	\$8.00	1	0	*	-1.21
26	131	\$6.45	\$8.00	1	0	*	*


* Own and cross-elasticity values can't be estimated due to lack of price changes during two consecutive weeks.

It was not possible to estimate own or cross elasticities for most of the weeks due to the absence of consecutive price changes. When estimation was possible, own elasticity values for six out of seven weeks ranged from -2.27 to -13.97 indicating that for those weeks product A was elastic. In week 3 price change was not accompanied by change in quantity as the result elasticity value was zero. For the same reason (lack of different prices in consecutive weeks) cross elasticities were estimated for only 10 of the 26 weeks.

Of these only four showed positive and the remaining six showed negative values. In the four weeks, when cross elasticities were positive, the two products were behaving as substitutes and in the remaining six weeks the two products were acting as complementary products. Although this appears intriguing, it simply shows the variability of price-demand relationships across weeks and also indicates the limitations of estimating elasticities using the mid-point formula estimation method as each value depends on only two weeks of data. Although the mid-point formula estimation method is easy to compute and may have some value in certain applications, such as price testing, where the objective is to see how price change affects demand for extended durations, it fails to capture the dynamics relationships of price and demand over an extended time frame (several weeks, months or years). In addition, various factors that influence demand such as demography, market share, promotion(s) and seasonality can't be part of the estimation process. For these main reasons, elasticity estimation using the regression approach offers a better alternative.

Product A and B had 8 and 10 distinct prices within six months indicating that the prices were changing about every three weeks. The presence of many distinct prices helps see how demand responds to various price points and is an essential requirement to estimate elasticities using the OLS regression. Figure 3 shows the sales and price change patterns of Product A, the price of Product B, and the two promotions of product A. Figure 3 is a graphical presentation of the same data displayed in Table 1. Overall as the price of Product A decreased the price of product B increased, and correspondingly the quantity of product A decreased.

Figure 3 Price and quantity of product A with price of product B during 26 weeks of sale.



When price elasticity of product A was estimated using only two variables, quantity and price, with the linear (Equation 4) and log-log model (Equation 5), virtually identical values of -3.777 and -3.795 (Table 2) were obtained. Note that elasticity value of the linear model follows that indicated in Equation 2; the slope (the unit change in quantity / unit change in price) was multiplied by the ratio of the average mean values of quantity and price. The two values indicate that customers are price sensitive, and a 1% price increase of product A will result in a 3.8% decrease in quantity of product A.

Table 2 Price elasticity values for product A computed by the linear and log-log model using only quantity and price.

<u>Parameter</u>	<u>Estimate</u>	<u>StdErr</u>	<u>tValue</u>	<u>Probt</u>	<u>mean_quantity</u>	<u>mean_price</u>	<u>elasticity</u>
price1	-54.804	5.309	-10.32	<.0001	99.038	6.859	-3.795
logp1	-3.776	0.343	-10.98	<.0001	.	.	.

It is believed that when an item has an elastic demand response, chances of having substitute items are high [5]. When the log-log model was run with covariates (promotion 1, promotion 2) and price of product B as shown in Equation 6, a better picture of how demand responses to price and other factors emerges (Table 3). First, the overall F statistics is significant at $p=0.01$ indicating that the model explains a significant portion of the variation in the data. The F value is also testing the hypothesis that all coefficients in the model, except the intercept, are equal to zero. The R^2 value, which explains the fraction of the total variation in the value of quantity due to the linear relationship with the list of covariates included, is high (0.91). The elasticity value of product A has changed from -3.795 (without covariates and price of product B) to -4.324.

Although the cross-elasticity value was 0.048, which is insignificant at $p=0.1$, its positive sign indicates that product B is a very marginal substitute to product A, and the value can be interpreted as a 1% increase in the price of product B can bring a 0.05% increase in the demand of product A. The relationship between the demand of product A and the prices of product A and B is shown in Figure 4. Overall while the demand of product A declined with the rise of its own price, its demand increased in response to price increase of product B.

The impact of promotion 1 (phone call) was significant at $p=0.1$ level. Price elasticity of product A with promotion 1 is -24.106, the summation of parameter estimates for log price of product A and promotion 1 (-4.321 + -19.7846). Price reduction coupled with phone calls to customers had the greatest impact on demand than price reduction alone. The second promotion (web ad) was not significant by itself at $p=0.1$, however, it showed some synergistic effect with price to improve elasticity value of product A (-6.632).


Table 3 Summary statistics of the log-log model including ANOVA, fit statistics and parameter estimates.

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	6	0.8439	0.1406	33.76	<.0001
Error	19	0.0791	0.0042		
Corrected Total	25	0.9230			

R-Square	Coeff Var	Root MSE	logq1 Mean
0.9142	1.4100	0.0645	4.5776

Parameter	Estimate	Standard Error	t Value	Pr > t
Intercept	12.8201	4.1475	3.09	0.0060
logp1	-4.3241	1.1880	-3.64	0.0017
logp2	0.0486	1.0971	0.04	0.9651
promotion1	36.8834	21.1735	1.74	0.0977
promotion2	4.5001	4.5482	0.99	0.3349
logp1*promotion1	-19.7846	11.3527	-1.74	0.0975
logp1*promotion2	-2.3087	2.3378	-0.99	0.3358

Figure 4 The relationships of product A and product B prices on quantity of product A.



There are rich theoretical and empirical evidences and various methods to establish the relationship between price and demand. The mid-point formula [5, 14], the linear model to

estimate the demand of beef using supermarket scanner data [7], the log-linear model to measure the determinants of elasticities of store and national brands of cheese [9] and several product categories [8], the log-log model to evaluate consumer good items [2], gasoline [10] and the Japanese timber market [17], the linear-log model to measure the demand response of alcoholic beverages [16] are the most common.

CONCLUSION

Some workers have made comparisons among alternative functional forms prior to selecting the best choice. Jaeweon et al. [11] have used the linear, multiplicative, attraction and Gutenberg models to analyze several brands and chose those that gave the best response function. Similarly Bolton [2] compared three alternative functional forms of sales equations (linear, multiplicative and exponential) to assess the robustness of the estimate and found that functional forms exhibit substantial differences. Therefore it is up to the marketing scientist to test various models and choose the one with the best fit. The analysis and interpretation described above and the attached SAS[®] code can be of great value to those who would like to measure item or store level elasticities using transaction data. The impact of marketing variables such as competition and customer demography [18], promotions [2], advertisement [4] and share of wallet [6] have also been incorporated to measure price elasticity of several category of products. Similar to such findings and as shown in this presentation, inclusion of covariates would improve parameter estimation and better decision making in product pricing and marketing.

REFERENCES

- [1] Bishwa B. Adhikari, Chen Zhen, Jennifer W. Kahende, Joshua Goetz, and Brett Loomis. 2012. Price Responsiveness of Cigarette demand in US: Retail Scanner Data (1994-2007). Economics Research International. Volume 2012 (2012), Article ID 148702: 1-11
- [2] Ruth Bolton. 1989. The Robustness of Retail-Level Price Elasticity Estimates. Journal of Retailing. 65:193-219.
- [3] Ronald W. Cotterill. 1994. Scanner Data: New Opportunities for Demand and Competitive Strategy Analysis. Agricultural and Resource Economics Review. 23:125-139.
- [4] T"u" lin Erdem, Michael P. Keane and Baohong Sun. 2008. The impact of advertising on consumer price sensitivity in experience goods markets. Quantitative Marketing and Economics. 6:139-176.Short-run E varied across time, and the ranges are given.

- [5] E. Genchev and J. Yarkova. 2010. Analysis of Price Elasticity of Food Products (For the Period 1999-2000). *Trakia Journal of Sciences*. 8:191-194.
- [6] Brett Gordon, Avi Goldfarb, and Yang Li. 2012. Does Price Elasticity Vary with Economic Growth? A Cross-category Analysis. *Journal of Marketing Research*. e-View Just Accepted.
- [7] Fred H. Hayes and Stephen A. DeLurgio. 2007. "Where is the Beef?": Statistical Demand Estimation Using Supermarket Scanner Data. *Journal of Case Research in Business and Economics*. Pages 1-11.
- [8] Stephan J. Hoch, Byung-Do Kim, Alan L. Montgomery, and Peter E. Rossi. 1995. Determinants of Store-Level Price Elasticity. *Journal of Marketing Research*. 32:17-29.
- [9] Min-Hsin Huang, David E. Hahn and Eugene Jones. 2004. Determinants of Price Elasticities for Store Brands and National Brands of Cheese. Paper presented at the American Agricultural Economics Association Annual Meeting, Denver, Colorado. August 1-4.
- [10] Jonathan E. Hughes, Christopher R. Knittel and Daniel Sperling. 2008. Evidence of a Shift in the Short-Run Price Elasticity of Gasoline Demand. *Energy Journal*. 2008, 1:113-134.
- [11] Hong Jaeweon, Cho Wanwoo, Jang Ho and Kwak Youngsik. 2010. Application Service Program (ASP) Price Elasticities. *International Journal of u- and e- Service, Science and Technology*. 3:23-30.
- [12] Alfred Marshall. 1890. *Principles of Economics*. Revised Edition, Macmillan, London, UK.
- [13] Andreas W. Neumann. 2007. Price Elasticity of Digital Scientific Information – A Field Experiment. *IADIS International Journal on WWW/Internet*. 5:1-11.
- [14] Gareth Wyn Owen. 2012. Applying Point Elasticity of Demand Principles to Optimal Pricing in Management Accounting. *The International Journal of Applied Economics and Finance*. 6:89-99.
- [15] Alan Robinson. 2007. A Review of Price Elasticity Models for Postal Products. Pitney Bowes Background Paper 2007-01.

- [16] Guoqiang Tian and Feng Liu. 2011. Is the Demand for Alcoholic Beverages in Developing Countries Sensitive to Price? Evidence from China. *International Journal of Environmental Research and Public Health*. 8:2124-2131.
- [17] Kiyoshi Yukutake, Takanori Saikai and Ashi Yohimoto. 1994. Modeling the Japanese Timber Market. Paper presented to Symposium on Systems Analysis in Forest Resources. Pacific Grove, California. September 6-9.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Please feel free to contact the authors at:

Dawit Mulugeta, Ph. D.
Manager, Pricing Analytics Team
Cardinal Health, Dublin, OH,
E-mail: dawit.mulugeta@cardinal.health.com

Jason Greenfield
Consultant, Pricing Analytics Team
Cardinal Health, Dublin, OH,
E-mail: jason.greenfield@cardinal.health.com

Tison Bolen
Senior Consultant, Pricing Analytics Team
Cardinal Health, Dublin, OH,
E-mail: tison.bolen@cardinal.health.com

Lisa Conley
Director, Pricing Analytics Team
Cardinal Health, Dublin, OH,
E-mail: lisa.conley@cardinal.health.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.

SAS® CODE

```

*****
** This Code Generates Elasticity and Cross-Elasticity Values Using **
** the Two-Point Estimation Method and the Log-Log OLS Regression for **
** Two Products that have 26 Weeks of Sales Data. **
***** ;

*****
** Generate log values of quantity and price, and compute Elasticity **
** and Cross-Elasticity values using the Two-Point Estimation Method. **
***** ;

data invoices;
  input week quantity1 price1 price2 promotion1 promotion2 ;
  format Price2 Price1 dollar10.2 ;
  logq1 = log(quantity1) ;
  logp1 = log(price1) ;
  logp2 = log(price2) ;
  lagp1 = lag(price1) ;
  lagp2 = lag(price2) ;
  lagq1 = lag(quantity1) ;
  if price1 - lagp1 ne 0
  then own = (mean(price1, lagp1)/mean(quantity1, lagq1))*((quantity1-
lagq1)/(price1-lagp1)) ;
  else own = . ;
  if price2 - lagp2 ne 0
  then cross = (mean(price2,lagp2)/mean(quantity1, lagq1))*((quantity1-
lagq1)/(price2-lagp2)) ;
  else cross = . ;
datalines;
1 70 7.50 7.00 0 0
2 75 7.30 7.15 0 0
3 75 7.25 7.15 0 0
4 78 7.25 7.15 0 0
5 100 6.50 7.20 1 0
6 120 6.50 7.20 1 0
7 95 6.50 7.25 1 0
8 115 6.50 7.25 1 0
9 78 7.05 7.25 0 0
10 85 7.05 7.25 0 1
11 88 7.05 7.30 0 1
12 84 7.05 7.30 0 1
13 90 7.05 7.30 0 1
14 86 7.05 7.40 0 0
15 95 7.00 7.40 0 0
16 88 7.00 7.40 0 0
17 105 6.90 7.40 0 0
18 100 6.90 7.45 0 1
19 108 6.90 7.45 0 1
20 95 6.90 7.50 0 1
21 98 6.90 7.50 0 1
22 130 6.45 7.50 1 0
23 125 6.45 7.70 1 0
24 133 6.45 7.75 1 0
25 128 6.45 8.00 1 0
26 131 6.45 8.00 1 0
;
run ;
proc format ;
  value undefmt . = '*' other = [8.2] ;

title 'Raw Data' ;

```

```

footnote "Note: '*' indicates that elasticity value cannot be estimated due to no
price change" ;
proc report data=invoices FORMCHAR='_' SPLIT='*' nowindows headline headskip ;
  column week quantity1 price1 price2 promotion1 promotion2 own cross;
  define week / '**Week' group center;
  define quantity1 / '*Product A*Quantity' center;
  define price1 / '*Product A*Price' center;
  define price2 / '*Product B*Price' center;
  define promotion1 / '*Product A*Promotion #1' center;
  define promotion2 / '*Product A*Promotion #2' center ;
  define own / 'Point*Elasticity*(own)' center format=undefmt.;
  define cross / 'Point*Elasticity*(cross)' center format=undefmt.;
run;
footnote;

%annomac;

data anno;
  %system(2, 2)
  %label(24,33,'|- Promo. 1 -|',*,0,0,1.25,duplex);
  %label(6.5,33,'| Promo. 1 |',*,0,0,1.25,duplex);
  %label(19.5,33,'| Promo. 2 |',*,0,0,1.25,duplex);
  %label(11.5,33,'| Promo. 2 |',*,0,0,1.25,duplex);
run;

*****
** Generate the Graph on Figure 3 that show the relationship between **
** prices and quantity of productA. **
***** ;
options reset=all ftext=SWISS htext=2.5 ;
  axis1 order=(25 to 150 by 25) minor=none label=(angle=90 'Quantity (units)' )
offset=(1) ;
  axis2 order=(6.25 to 8.25 by .5) minor=(number=1) label=(angle=90 'Price ($)')
offset=(1);
  axis3 order=(0 to 27 by 3) minor=(number=2) label=('Week') offset=(1) ;
  symbol1 color=black i=join value=circle height=2 width=2 ;
  symbol2 color=black i=join value=square height=2 width=2 ;
  symbol3 color=black i=join value=triangle height=2 width=2 ;
  legend1 label=none mode=reserve position=(top center outside) value=('Quantity
A' ) shape=symbol(5,1) ;
  legend2 label=none mode=reserve position=(top center outside) value=('Price A'
'Price B') shape=symbol(3,1) ;
proc gplot data=invoices ;
  plot quantity1*week=1 /overlay annotate=anno legend=legend1 vaxis=axis1
haxis=axis3 ;
  plot2 price1*week=2 price2*week=3 /overlay legend=legend2 vaxis=axis2 ;
run ;

*****
** Generate the Graph on Figure 4 that show the price and quantity of **
** Product A with price of product B. **
***** ;
proc reg data=invoices noprint ;
  model quantity1=price1 ;
  output out=yhat1 predicted=yhat1 ;
run ;
proc reg data=invoices noprint ;
  model quantity1=price2 ;
  output out=yhat2 predicted=yhat2 ;
run ;
proc sql;

```

```

create table forplot as
  select a.week,
         a.quantity1,
         a.pricel,
         yhat1,
         a.price2,
         yhat2
  from invoices a, yhat1 b, yhat2 c
  where a.week = b.week = c.week ;
quit ;
goptions reset=all ftext=SWISS htext=2.5 ;
  axis1 order=(0 to 150 by 25) minor=none label=(angle=90 'Quantity of A' )
offset=(1) ;
  axis2 order=(6.25 to 8.25 by .5) minor=(number=1) label=('Price ($)')
offset=(5) ;
  symbol1 color=black value=circle height=2 width=2 ;
  symbol2 color=black value=square height=2 width=2 ;
  symbol3 color=black i=join line=7 height=2 width=2 ;
  symbol4 color=black i=join line=2 height=2 width=2 ;
  legend1 label=none position=(top center outside) value=('Product A' 'Product
B') shape=symbol(5,1) ;
proc gplot data=forplot ;
  plot quantity1*pricel=1 quantity1*price2=2 yhat1*pricel=3 yhat2*price2=4 /overlay
legend=legend1 vaxis=axis1 haxis=axis2 ;
run ;

*****
** Estimate elasticity based on only quantity and price using linear **
** regression model for product A. **
***** ;
ods listing close;
proc glm data=invoices ;
  model quantity1 = pricel ;
  ods output
    ParameterEstimates = pe1 ;
run ;
ods listing;

proc sql noprint;
  select mean (quantity1) into: mean_quantity1 from invoices ;
  select mean (pricel) into: mean_pricel from invoices ;
quit ;

data pe1 ;
set pe1 ;
  mean_quantity = &mean_quantity1 ;
  mean_price = &mean_pricel ;
  if parameter ne "Intercept" then elasticity = estimate * (&mean_pricel /
&mean_quantity1) ;
run ;

*****
** Estimate elasticity based on only quantity and price using log-log **
** regression model for product A. **
***** ;
ods listing close;
proc glm data=invoices;
  model logq1 = logp1 ;
  ods output
    ParameterEstimates = pe2 ;
run;
ods listing;

```

```

data pe;
  set pe1 pe2;
run;

proc print data=pe noobs;
  where parameter ne 'Intercept';
  var Parameter Estimate StdErr tValue Probt mean_quantity mean_price elasticity;
run;

*****
** Estimate cross-elasticity using linear regression model for product A. **
***** ;
ods listing close;
proc glm data=invoices ;
  model quantity1 = price1 price2 ;
  ods output
    ParameterEstimates = cel ;
run ;
ods listing;

proc sql noprint;
  select mean (price2) into: mean_price2 from invoices ;
quit ;

data cel ;
set cel ;

  if parameter eq "price2" then do;
    mean_quantity = &mean_quantity1 ;
    mean_price = &mean_price2 ;
    elasticity = estimate * (&mean_price2 / &mean_quantity1) ;
  end;
run ;

*****
** Estimate cross-elasticity using log-log regression model for product A. **
***** ;
ods listing close;
proc glm data=invoices;
  model logq1 = logp1 logp2 ;
  ods output
    ParameterEstimates = ce2 ;
run;
ods listing;

data ce;
  set cel ce2;
run;

proc print data=ce noobs;
  where parameter ne 'Intercept';
  var Parameter Estimate StdErr tValue Probt mean_quantity mean_price elasticity;
run;

*****
** Estimate elasticity based on quantity, price, promotions of product**
** A and price of product B using log-log regression model. **
***** ;
ods listing close;
proc glm data=invoices ;
  model logq1 = logp1 logp2 promotion1 promotion2 logp1*promotion1 logp1*promotion2
;
  ods output

```



```
OverallANOVA = OverallANOVA3  
FitStatistics = FitStatistics3  
ParameterEstimates = pe3 ;  
  
run ;  
ods listing;  
  
proc print data=OverallANOVA3 noobs ;  
proc print data=FitStatistics3 noobs;  
proc print data=pe3 noobs;  
run ;
```